21/replication_package/Do_Files/Figures/B4_graph_vouchersXbins.do

preserve
	
collapse (mean) treat_scheduled, by( cscomnum date_morning_survey )
sort cscomnum date_morning_survey
by cscomnum: gen bins=_n

gen bin1=0
replace bin1=1 if bins==1 | bins==2
gen bin2=0
replace bin2=1 if bins==3 | bins==4
gen bin3=0
replace bin3=1 if bins==5 | bins==6

keep cscomnum date_morning_survey bin1 bin2 bin3 bins
sort cscomnum date_morning_survey

* Save this file for the merge
tempfile auxiliar
sa `auxiliar'

restore

sort cscomnum date_morning_survey

merge m:1 cscomnum date_morning_survey using `auxiliar'

assert _m==3
drop _m

* create variables that have the interaction dummies, say, called PV_day1, DV_day1, PV_day2, … etc. 
forvalues i=1/3 {

gen PV_bin`i'=patient_voucher*bin`i'
gen DV_bin`i'=doctor_voucher*bin`i'
}

* generate a set of variables from PV:  V_day1 = PV_day1, V_day2= PV_day2 etc.
forvalues i=1/3 { 

rename PV_bin`i' V_bin`i'

}

foreach var in used_vouchers_admin RXtreat_sev_simple_mal  RXtreat_severe_mal treat_severe_mal treat_sev_simple_mal {

pdslasso `var' DV_bin*  V_bin*  ($date $patient bin1 bin2), clu($cluvar) partial($date bin1 bin2 )

est sto PV`var'
}

forvalues i=1/3 {

rename V_bin`i' PV_bin`i'

}
*generate a set of variables this time from DV: V_day1 = DV_day1, V_day2= DV_day2 etc.
forvalues i=1/3 {

rename DV*bin`i' V_bin`i'

}
	
foreach var in used_vouchers_admin RXtreat_sev_simple_mal  RXtreat_severe_mal treat_severe_mal treat_sev_simple_mal {

pdslasso `var' PV_bin*  V_bin*  ($date $patient bin1 bin2), clu($cluvar) partial($date bin1 bin2)

est sto DV`var'
}
drop V_bin*	
	
coefplot (PVused_vouchers_admin, keep(V_bin* ) label(Patient Voucher) lpat(solid) lcol(black) lwid(medthick)) (DVused_vouchers_admin, keep(V_bin*) label(Doctor Voucher) lpat(longdash) lcol(gs10) lwid(medthick))  ///
  , vertical legend(size(small)) graphregion(color(white)) xtit("Clinic Observation Day", size(small)) ///
  title("A. Used Vouchers", size(small)) recast(line)   ///
   name(PE, replace) xlab(1 "1-2" 2 "3-4" 3 "5-6", labsize(small)) scheme(s2mono)
   
coefplot (PVRXtreat_sev_simple_mal, keep(V_bin* ) label(Patient Voucher) lpat(solid) lcol(black) lwid(medthick)) (DVRXtreat_sev_simple_mal, keep(V_bin*) label(Doctor Voucher) lpat(longdash) lcol(gs10) lwid(medthick))  ///
  , vertical legend(size(small)) graphregion(color(white)) xtit("Clinic Observation Day", size(small)) ///
  title("B. Prescribed Any Malaria Treatment", size(small)) recast(line) ///
   name(PA, replace) xlab(1 "1-2" 2 "3-4" 3 "5-6", labsize(small)) scheme(s2mono)
   
coefplot (PVtreat_sev_simple_mal, keep(V_bin* ) label(Patient Voucher) lpat(solid) lcol(black) lwid(medthick)) (DVtreat_sev_simple_mal, keep(V_bin*) label(Doctor Voucher) lpat(longdash) lcol(gs10) lwid(medthick))  ///
  , vertical legend(size(small)) graphregion(color(white)) xtit("Clinic Observation Day", size(small)) ///
  title("C. Purchased Any Malaria Treatment", size(small)) recast(line) ///
   name(PB, replace) xlab(1 "1-2" 2 "3-4" 3 "5-6", labsize(small)) scheme(s2mono)
   
coefplot (PVRXtreat_severe_mal, keep(V_bin* ) label(Patient Voucher) lpat(solid) lcol(black) lwid(medthick)) (DVRXtreat_severe_mal, keep(V_bin*) label(Doctor Voucher) lpat(longdash) lcol(gs10) lwid(medthick))  ///
  , vertical legend(size(small)) graphregion(color(white)) xtit("Clinic Observation Day", size(small))  ///
  title("D. Prescribed Severe Malaria Treatment", size(small)) recast(line) ///
   name(PC, replace) xlab(1 "1-2" 2 "3-4" 3 "5-6", labsize(small)) scheme(s2mono)
   
coefplot (PVtreat_severe_mal, keep(V_bin* ) label(Patient Voucher) lpat(solid) lcol(black) lwid(medthick)) (DVtreat_severe_mal, keep(V_bin*) label(Doctor Voucher) lpat(longdash) lcol(gs10) lwid(medthick))  ///
  , vertical legend(size(small)) graphregion(color(white)) xtit("Clinic Observation Day", size(small)) ///
  title("E. Purchased Severe Malaria Treatment", size(small)) recast(line) ///
   name(PD, replace) xlab(1 "1-2" 2 "3-4" 3 "5-6", labsize(small)) scheme(s2mono)  
   
grc1leg PE PA PB PC PD , row(3) graphregion(color(white)) name(final, replace)  //xsize(9) ysize(13) 
	gr dis final, xsize(9) ysize(13)
	
 
graph export "$graphs/B4_vouchersXbins.pdf", replace